

MOVT終了時の EN_DACL(減速有効) タイミング

ACCEL ti2k-181023 (LastUpdate 18/10/23)


使用ボード : MPC-2200X BL/I 1.14_71 2018/08/31 with MPG-2314
 ユーザーズマニュアル コマンドリファレンス MOVT 参照

■直線移動で終わる場合

EN_DACLが WAIT RR(axis)==0 の前

```
FOR pnt=1001 TO X(2000)
  MOVT axis P(pnt) X(1000+pnt)
  linecolor=pnt%3+1
NEXT
```


```
EN_DACL
WAIT RR(axis)==0
/*EN_DACL
```





EN_DACLが WAIT RR(axis)==0 の後

```
FOR pnt=1001 TO X(2000)
  MOVT axis P(pnt) X(1000+pnt)
  linecolor=pnt%3+1
NEXT
```

```
/*EN_DACL
WAIT RR(axis)==0
EN_DACL
```



サンプリングの都合で多少斜めになっていますが減速域は有りません



■円弧補間で終わる場合

EN_DACL は直線移動に有効です。
円弧補間で終了する場合は減速域はありません。

EN_DACLが WAIT RR(axis)==0 の前




```
FOR pnt=1001 TO X(2000)-1
  MOVT axis P(pnt) X(1000+pnt)
  linecolor=pnt%3+1
NEXT
```

```
EN_DACL
WAIT RR(axis)==0
/*EN_DACL
```

EN_DACLが WAIT RR(axis)==0 の後

```
FOR pnt=1001 TO X(2000)-1
  MOVT axis P(pnt) X(1000+pnt)
  linecolor=pnt%3+1
NEXT
```

```
/*EN_DACL
WAIT RR(axis)==0
EN_DACL
```



■プログラム全部

```

blue=1
red=2
green=3

MEWNET 115200 2          /* 通信宣言
pgnum=0
QUIT_FORK 1 *main
END

*main
FILL P(10000) 10000 0    /* 点データ初期化
PG pgnum
CLRPOS
axis=X_A|Y_A
ACCEL ALL_A 50000

GOSUB *SET_POINT        /* 作業点と移動形態の作成

QUIT_FORK 2 *PgRecord   /* 記録開始
TIME 50

linecolor=3
MOVL P(1000)            /* スタートポイントへ移動 (3軸移動)
WAIT RR(ALL_A)==0
TIME 10
FEED axis Y(2000)      /* スピード変更
DS_DACL

FOR pnt=1001 TO X(2000) /* P(1001)~P(1004)連続移動
  MOVT axis P(pnt) X(1000+pnt) /* P(pnt)は移動先、X(1000+pnt)は円弧または直線指定
  linecolor=pnt%3+1
NEXT

EN_DACL
WAIT RR(axis)==0
/*EN_DACL

TIME 50
QUIT 2                  /* 記録終了
PR "Oshimai"
END

*SET_POINT
SETP 1000 10000 20000 0 -10000 /* スタートポイント
SETP 1001 30000 20000 20000 20000 /* XY=円弧終点座標、UZ=円弧中心座標
SETP 1002 30000 10000 0 0 /* XY=移動先座標
SETP 1003 10000 10000 20000 10000 /* XY=円弧終点座標、UZ=円弧中心座標
SETP 1004 10000 20000 0 0 /* XY=移動先座標
SETP 2000 1004 50 0 0 /* X=最終点番号、Y=FEED
SETP 2001 CW 0 0 0 /* P(1000) から P(1001) は CW
SETP 2002 0 0 0 0 /* P(1001) から P(1002) は 直線
SETP 2003 CW 0 0 0 /* P(1002) から P(1003) は CW
SETP 2004 0 0 0 0 /* P(1003) から P(1004) は 直線
RETURN

*PgRecord                /* 記録タスク mod 181023
PG pgnum
FILL P(10000) 10000 0
IF linecolor==0 THEN : linecolor=1 : END_IF
oldposX_=X(0)
oldposY_=Y(0)
oldposZ_=Z(0)
pointnum_=10000          /* 加減速記録先頭点番号
locusnum_=15000          /* 軌跡記録先頭点番号
strclk_=SYSCLK
DO
  nowclk_=SYSCLK-strclk_ /* 経過時間
  nowposX_=X(0)
  nowposY_=Y(0)
  nowposZ_=Z(0) /* 現在位置
  /* 点番号 色 X現位置 Y現位置 Z現位置
  SETP locusnum_ linecolor nowposX_ nowposY_ nowposZ_
  locusnum_=locusnum_+1
  IF locusnum_%4==0 THEN
    difposX_=nowposX_-oldposX_ /* X差分
    oldposX_=nowposX_
    difposY_=nowposY_-oldposY_ /* Y差分
    oldposY_=nowposY_
    difposZ_=nowposZ_-oldposZ_ /* Z差分
    oldposZ_=nowposZ_
    /* 点番号 時間 X差分 Y差分 Z差分
    SETP pointnum_ nowclk_ difposX_ difposY_ difposZ_
    pointnum_=pointnum_+1
  END_IF
  TIME 2 /* 記録間隔msec
LOOP

```